Project AMPL

Όνομα: Παππάς Δημήτριος

ΑΕΜ: 8391

Email: dspappas@ece.auth.gr

Καθηγητής: Παπαλάμπρου Κ.

Πρόβλημα

Το πρόβλημα που κλήθηκα να αντιμετωπίσω αφορά, την μοντελοποίηση τεσσάρων οργχείων, εκ των οποίων μόνα τα τρία ή λιγότερα μπορούν να είναι ταυτόχρονα ανοιχτά για κάθε έτος. Τα ορυχεία πληρώνουν κάθε έτος εισφορές, ακόμα και στην περίπτωση που μένουν κλειστά για εκείνο το έτος, όμως επιθυμούν να λειτουγήσουν κάποιο από τα επόμενα. Υπάρχει η δυνατότητα να κλείσει οριστικά κάποιο ορυχείο, ώστε να μην πληρώνει τις εισφορές. Η ανάλυση γίνεται για τέσσαρα ορυχεία στη διάρκεια πέντε ετών. Κάθε έτος το μίγμα των ορυκτών θα πρέπει να ικανοποιεί έναν συγκεκριμένο δείκτη ποιότητας.

Δεδομένα

Ορυχεία	Εισφορές(10^6 €)	Άνω Όριο Εξόρυξης(10^6 $tons$)	Δείκτης Ποιότητας
1	5	2	1
2	4	2,5	0,7
3	4	1,3	1,5
4	5	3	0,5

Έτη	Δείκτης Ποιότητας Μίγματος
1	0,9
2	0,8
3	1,2
4	0,6
5	1

<u>Έσοδα:</u> Τελικό μίγμα εξόρυξης επί 10€/ton.

<u>Έξοδα:</u> Οι εισφορές του παραπάνω πίνακα.

Τα έσοδα και τα έξοδα προεξοφλούνται με ετήσιο επιτόκιο 10%.

Μοντελοποίηση

Το πρόβλημα αντιμετοπίστηκε με το μοντέλο του ακέραιου γραμμικού προγραμματισμού αρχικά σε θεωρητική ανάλυση και στη συνέχεια με την βοήθεια του προγράμματος AMPL. Στόχος μας ήταν η μεγιστοποίηση του κέρδους.

<u>Μεταβλητές Απόφασης:</u>

Ορίζω τρεις πίνακες x,y,z, με διαστάσεις 4x5.
Ο x περιλαμβάνει την ποσότητα εξόρυξης, με τιμές $x_{i,j}$, για το ορυχείο i το έτος j.
Ο y είναι Boolean (δυαδική μεταβλητή απόφασης) και παίρνει τιμή “1”, όταν το ορυχείο i λειτουργεί το έτος j και “0” όταν δεν λειτουργεί (προσωρινά).
Ο z είναι κι αυτός Boolean και παίρνει τιμή “1”, όταν το ορυχείο i είναι ανοιχτό το έτος j και “0” όταν είναι κλειστό μόνιμα (έχει βάλει “λουκέτο”).

```
# Decision Variables
var x {i in Mines, j in Years} >=0; # Ποσότητα εξόρυξης του ορυχείου i το j έτος
var y {i in Mines, j in Years}  binary;     # Το ορυχείο i λειτουργεί/δεν λειτουργεί το j έτος
var z {i in Mines, j in Years}  binary;     # Το ορυχείο i είναι ανοιχτό/κλειστό (μόνιμα) το j έτος
```

<u>Παράμετροι:</u>

Οι παράμετροι που έχουν οριστεί αντιστοιχούν στους παραπάνω πίνακες δεδομένων και είναι όλοι μίας διάστασης.

```
# Parameters
param mining_limit {Mines} >=0;          # Άνω όριο εξόρυξης για κάθε ορυχείο
param contributions {Mines} >=0;         # Εισφορές κάθε ορυχείου
param quality {Mines} >=0;               # Δείκτης Ποιότητας για κάθε ορυχείο
param mixture_quality {Years} >=0;       # Δείκτης Ποιότητας Μίγματος για κάθε έτος
```

Περιορισμοί:

Για την επίλυση του προβλήματος, χρειάστηκαν να υλοποιηθούν πέντε περιορισμοί.

Mining_Limit:

```
# Restrictions
s.t. Mining_Limit {i in Mines, j in Years}:
    x[i,j] <= mining_limit[i] * y[i,j];
```

Με την ανίσωση αυτή πετυχαίνω άνω όριο εξόρυξης ίσο με την παράμετρο mining_limit[i] , όταν το y[i,j] είναι 1. Αντίθετα, όταν y[i,j]=0, δηλαδή το ορυχείο i δεν θα λειτουγήσει το j έτος, τότε εξαναγκάζω και την ποσότητα εξόρυξης x να είναι 0.

Working_Mines_Each_Year:

```
s.t. Working_Mines_Each_Year {j in Years}:
    sum {i in Mines} y[i,j] <= 3;
```

Κάθε έτος μπορούν να λειτουργούν μέχρι και τρία ορυχεία.

Mixture_Quality_Each_Year:

```
s.t. Mixture_Quality_Each_Year {j in Years}:
    sum {i in Mines} (quality[i] * x[i,j]) = mixture_quality[j] * sum {k in Mines}  x[k,j];
```

Για κάθε έτος, αθροίζω τις ποσότητες εξόρυξης, που έχουν ως συντελεστή τον δείκτη ποιότητας quality[i], και τις εξισώνω με τον δείκτη μίγματος πολλαπλασιασμένο με το άθροισμα των ποσοτήτων εξόρυξης. Λόγω του περιορισμού **Mining_Limit,** εξασφαλίζω την ύπαρξη μηδενικού στα x τα οποία δεν λειτουργουν (y=0) και έτσι δεν χρειάζεται να πολλαπλασιάσω x*y που με οδηγεί σε μη γραμμικότητα.

*** Να σημειωθεί ότι ο συγκεκτημένος περιορισμός μου προκάλεσε πρόβλημα μη γραμμικότητας, διότι το αθροισμα των x το είχα στον παρονομαστή του $1^{ου}$ όρου και ως αποτέλεσμα έπαιρνα κέρδος 0. Το προβλημα λύθηκε με την μεταφορά του στο δεξιό μέλος της εξίσωσης και την ύπαρξη γραμμικότητας. ***

Closed1 & Closed2:

```
s.t. Closed1 {i in Mines, j in Years}:
    z[i,j] >= y[i,j];

s.t. Closed2 {i in Mines, j in 1..t-1}:
    z[i,j] >= z[i,j+1];
```

Οι δύο παραπάνω περιορισμοί αφορούν το αν το ορυχείο θα είναι ανοιχτό ή κλειστό (μόνιμα).
Στην περίπτωση της **Closed1**, αν το ορυχείο είναι κλειστό z=0, τότε αναγκαστηκά δεν λειτουργεί y=0. Σε αντίθετη περίπτωση, αν είναι ανοιχτό z=1, τότε μπορεί είτε να λειτουργεί είτε να μη λειτουργεί για εκείνο το έτος y={0,1}.
Επίσης, με την **Closed2** εξασφαλίζουμε ότι αν την χρονιά j αποφασίσει να κλείσει το ορυχείο i (z[i,j]=0), τότε και την επόμενη χρονιά j+1 θα παραμείνει κλειστό (z[i,j+1]=0). Ενώ αν είναι ανοιχτό (z[i,j]=1), τότε την επόμενη χρονιά μπορεί να παραμείνει ανοιχτό ή να αποφασίσει να κλείσει(z[i,j+1]={0,1}). (Δουλεύω με την μέθοδο της αναδρομής. Κάθε τιμή εξαρτάται από την προηγούμενη).

<u>Αντικειμενική Συνάρτηση:</u>

```
# Objective Function
maximize profit:
    sum {j in Years} 1/(1.1)^(j-1) * (sum{i in Mines} (10*x[i,j] - contributions[i] * z[i,j]));
```

Η Αντικειμενική Συνάρτηση αποτελείται από την διαφορά των εξόδων/εισφορών από τα έσοδα όλων των ορυχείων για κάθε έτος, πολλαπλασιασμένη με τον συντελεστή $\frac{1}{(1+0,1)^{j-1}}$, όπου 0,1 είναι το επιτόκιο και j το έτος. Πολλαπλασιάζουμε τις εισφορές με την Boolean z, γιατί δεν πληρώνουμε εισφορές στην περίπτωση που το ορυχείο είναι κλειστό (μόνιμα).

Αποτελέσματα:

```
profit = 146862000

:        x      y   z    :=
1 1     2e+06   1   1
1 2         0   0   1
1 3   1950000   1   1
1 4    125000   1   1
1 5     2e+06   1   1
2 1         0   0   1
2 2   2500000   1   1
2 3         0   1   1
2 4   2500000   1   1
2 5   2166670   1   1
3 1   1300000   1   1
3 2   1300000   1   1
3 3   1300000   1   1
3 4         0   0   1
3 5   1300000   1   1
4 1   2450000   1   1
4 2   2200000   1   1
4 3         0   0   1
4 4     3e+06   1   1
4 5         0   0   0
;
```

Το συνολικό κέρδος είναι 146862000 = 146,862 εκατομμήρια €.

Το μοναδικό ορυχείο που κλείνει (μόνιμα) είναι το 4^{o} στο 5^{o} έτος.

Τα ορυχεία που λειτουργούν κάθε έτος y, καθώς και η ποσότητα ορυκτου που παράγουν x, φαίνονται στον παραπάνω πίνακα.